

Tema 12

Introducció a les Equacions Diferencials Ordinàries

Objectius:

1. Definir famílies de corbes planes amb wxMaxima. Representació gràfica.
2. Aprendre la sintaxi d'equacions diferencials amb wxMaxima.
3. Aprendre a resoldre algunes equacions diferencials de primer ordre. Tipus de solucions.
4. Aprendre a resoldre algunes equacions diferencials de segon ordre. Tipus de solucions.
5. Estudiar algunes aplicacions de les equacions diferencials
6. Estudiar la resolució numèrica d'equacions diferencials.

Continguts:

- 12-1. Famílies de corbes planes. Equació diferencial d'una família de corbes.
- 12-2. Equacions diferencials ordinàries d'ordre 1. Sistemes.
- 12-3. Equacions diferencials ordinàries d'ordre 2.
- 12-4. Introducció als mètodes numèrics de resolució d'equacions diferencials ordinàries i sistemes.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.
- ABD91 A. AUBANELL, A. BENSENY, A. DELSHAMS (1991)
Eines bàsiques de Càlcul Numèric

- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones
- GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- SI93 George F. SIMMONS (1993)
Ecuaciones Diferenciales.
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima
- ZI88 ZILL, Daniel G. (1988)
Ecuaciones Diferenciales con aplicaciones (2ª edición).

12-1.- Famílies de corbes planes. Equació diferencial d'una família de corbes.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_12-1.wxm**.

Una família uniparamètrica de corbes planes és un conjunt de corbes en el pla que es poden representar en forma implícita per una equació del tipus:

$$F(t, y, C) = 0, \quad C \in \mathbb{R}$$

on és C un paràmetre (és a dir un nombre qualsevol), t és la variable independent i y és la variable dependent. En ocasions l'equació anterior permet expressar l'equació en forma explícita:


$$y = f(t, C), \quad C \in \mathbb{R}$$

Exemple 12.1.1. El conjunt de corbes donat per l'equació $y = Ct$, $C \in \mathbb{R}$ és una família uniparamètrica de corbes planes (rectes que passen per l'origen). Vegem la seva definició i representació gràfica amb wxMaxima.

```
(%i1) f11(x):= 0.5*x$      f12(x):=x$      f13(x):=3*x$
      f14(x):=-0.5*x$     f15(x):=-x$     f16(x):=-3*x$

(%i7) wxplot2d([f11(x),f12(x),f13(x),f14(x),f15(x),f16(x)], [x,-2,2],
[style, [lines, 2]], [legend, ""],
[gnuplot_preamble, "set size ratio 1; set grid;"], [nticks, 300] )$
```

(%t7)




Exemple 12.1.2. El conjunt de corbes donat per l'equació $y = Ct^2$, $C \in \mathbb{R}$ és una família uniparamètrica de corbes planes (paràboles que passen per l'origen). Vegem la seva definició i representació gràfica amb wxMaxima.

```
(%i1) f31(x):=x^2$      f32(x):=2*x^2$      f33(x):=3*x^2$
      f34(x):=-x^2$     f35(x):=-2*x^2$     f36(x):=-3*x^2$

(%i7) wxplot2d([f31(x),f32(x),f33(x),f34(x),f35(x),f36(x)], [x,-2,2],
[style, [lines, 2]] , [legend, ""],
[gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 300] )$
```

(%t7)




Exemple 12.1.3. El conjunt de corbes donat per l'equació $y = e^{at}$, $a \in \mathbb{R}$ és una família uniparamètrica de corbes planes (funcions exponencials). Vegem la seva definició i representació gràfica amb wxMaxima.

```
(%i1) f11(x):=exp(0.5*x)$  f12(x):=exp(x)$
      f13(x):=exp(1.5*x)$  f14(x):=exp(2*x)$
      f21(x):=exp(-0.5*x)$ f22(x):=exp(-x)$
      f23(x):=exp(-1.5*x)$ f24(x):=exp(-2*x)$
```

```
(%i9) wxplot2d([f11(x),f12(x),f13(x),f14(x),f21(x),f22(x),f23(x),f24(x)], [x,-1,1],
[style, [lines, 2]] , [legend, ""],
[gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 300] )$
```

(%t9)




Exemple 12.1.4. El conjunt de corbes donat per l'equació $y = 2 - e^{at}$, $a \in \mathbb{R}$ és una família uniparamètrica de corbes planes (funcions exponencials). Vegem la seva definició i representació gràfica amb wxMaxima.

```
(%i1) f11(x):=2 - exp(- 0.5*x)$    f12(x):=2 - exp(- 1.0*x)$
      f13(x):=2 - exp(- 0.8*x)$    f14(x):=2 - exp(- 1.5*x)$
```

```
(%i5) wxplot2d([f11(x),f12(x),f13(x),f14(x)], [x, 0, 8], [y, 1.0, 2.2],
[style, [lines, 2]] , [legend, ""],
[gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 10] )$
```

(%t5)



Una família biparamètrica de corbes planes és un conjunt de corbes en el pla que es poden representar en forma implícita per una equació del tipus:

$$F(t, y, C_1, C_2) = 0, \quad C_1, C_2 \in \mathbb{R}$$

on $C_1, C_2 \in \mathbb{R}$ són els paràmetres (és a dir nombres qualssevol), t és la variable independent i y és la variable dependent. En ocasions l'equació anterior permet expressar l'equació en forma explícita:

$$y = f(t, C_1, C_2), \quad C_1, C_2 \in \mathbb{R}$$

Exemple 12.1.5. El conjunt de corbes donat per l'equació


$$y = C_1 \cos(2t) + C_2 \sin(2t), \quad C_1, C_2 \in \mathbb{R}$$

és una família biparamètrica de corbes planes (funcions trigonomètriques). Vegem la seva definició i representació gràfica amb wxMaxima.

```
(%i1) f51(t):=cos(2*t) + sin(2*t)$      f52(t):=1.5*cos(2*t) + 1.5*sin(2*t)$
      f53(t):=cos(2*t) - sin(2*t)$      f54(t):=1.5*cos(2*t) - 1.5*sin(2*t)$
```

```
(%i5) wxplot2d([f51(t),f52(t),f53(t),f54(t)], [t,-4,4],
               [style, [lines, 2]] , [legend, ""],
               [gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 10] )$
```

(%t5)



Equació diferencial d'una família de corbes planes.

Comencem pel cas d'una família uniparamètrica. Si en l'equació de la família $y = Ct^2$, $C \in \mathbb{R}$ es calcula la derivada considerant que $y = y(t)$ resulta:

$$y'(t) = 2Ct$$

Ara es pot eliminar el paràmetre amb les dues equacions: la de la família i la seva derivada, obtenint-se:

$$y'(t) = 2\left(\frac{y}{t^2}\right)t \Rightarrow y'(t) = \frac{2y}{t} \Rightarrow 2y - ty' = 0$$

L'última equació no conté paràmetres i la compleixen totes les corbes de la família. Aquesta equació, anomenada equació diferencial de la família de corbes planes, conté només la variable, la funció i la derivada de la funció. Aquesta equació s'anomena equació diferencial ordinària de primer ordre (o d'ordre 1).

Amb wxMaxima es procedeix de la manera següent. En primer lloc s'escriu la dependència funcional que s'assigna a les variables de l'equació:

```
(%i1) depends ([y], [t]);  
(%o1) [y(t)]
```

A continuació s'escriu l'equació de la família de corbes planes i es deriva l'equació:

```
(%i2) Eq1: y=C*t^2;  
      Eq2: diff(Eq1,t) ;  
(%o2) y = t^2 C  
(%o3)  $\frac{d}{dt}y = 2tC$ 
```

A continuació s'elimina el paràmetre entre aquestes dues equacions amb la instrucció "eliminate", que s'aplica amb la sintaxi que es veu tot seguit:

```
(%i4) eliminate([Eq1, Eq2], [C]);  
(%o4) [t(2y - t( $\frac{d}{dt}y$ ))]
```

Finalment es pot escriure l'equació diferencial de la família de corbes:

```
(%i5) edo1: 2*y-t*'diff(y,t) = 0;  
(%o5) 2y - t( $\frac{d}{dt}y$ ) = 0
```

En el cas d'una família biparamètrica, com per exemple

$$y = C_1 \cos(2t) + C_2 \sin(2t), \quad C_1, C_2 \in \mathbb{R}$$

es calcula la derivada considerant que $y = y(t)$ resulta:

$$y'(t) = -2C_1 \sin(2t) + 2C_2 \cos(2t), \quad C_1, C_2 \in \mathbb{R}$$

Si es torna a derivar en l'equació anterior, s'obté:

$$y''(t) = -4C_1 \cos(2t) - 4C_2 \sin(2t), \quad C_1, C_2 \in \mathbb{R}$$

Ara es poden eliminar els paràmetres amb les tres equacions: la de la família, la seva derivada i la derivada segona, obtenint-se:

$$y'' + 4y = 0$$

Aquesta equació no conté paràmetres, la compleixen totes les corbes de la família i s'anomena equació diferencial de la família de corbes planes; conté només la variable, la funció i/o les derivades de primer i segon ordre de la funció. Aquesta equació s'anomena equació diferencial ordinària de segon ordre (o d'ordre 2).

Amb wxMaxima es procedeix com abans, però ara es tracta d'eliminar les dues constants. Vegem el procediment:

```
(%i1) depends ([y] , [t]);
(%o1) [y(t)]

(%i2) Eq1: y = C1*cos(2*t) + C2*sin(2*t);
      Eq2: diff(Eq1, t) ;
      Eq3: diff(Eq2, t) ;

(%o2) y = sin(2 t) C2 + cos(2 t) C1

(%o3)  $\frac{d}{dt} y = 2 \cos(2 t) C2 - 2 \sin(2 t) C1$ 

(%o4)  $\frac{d^2}{dt^2} y = -4 \sin(2 t) C2 - 4 \cos(2 t) C1$ 

(%i5) eliminate([Eq1, Eq2, Eq3], [C1,C2]);
(%o5)  $[-\cos(2 t) \left( \frac{d^2}{dt^2} y + 4 y \right)]$ 

(%i6) edo2: 'diff(y,t,2)+4*y=0;
(%o6)  $\frac{d^2}{dt^2} y + 4 y = 0$ 
```


En relació a l'equació diferencial, la família de corbes planes s'anomena solució general de l'equació diferencial i cada una de les corbes de la família s'anomena una solució particular.

En una família uniparamètrica, per calcular una solució particular cal conèixer una condició sobre la corba, anomenada condició inicial, expressada en general en la forma $y(0) = y_0$ o bé $y(t_0) = y_0$; aquesta condició permet calcular el valor del paràmetre i determinar la solució particular corresponent. Per exemple en la família $y = Ct^2$ si la condició inicial és $y(2) = 5$ substituint aquesta condició en la solució general s'obté $C = \frac{5}{4}$ i, per tant, la solució particular buscada és $y = \frac{5}{4}t^2$.

Càlculs amb wxMaxima:

```
(%i1) y(t):=C*t^2;
      Eq1: y(2)=5;

(%o1) y(t):= C t^2
      (%o2) 4 C = 5

(%i3) solve(Eq1, C);

(%o3) [C = 5/4]

(%i4) C=C:5/4;

(%o4) C = 5/4

(%i5) y=C*t^2;

(%o5) y = 5 t^2/4
```

En una família biparamètrica, per calcular una solució particular cal conèixer dues condicions: una condició sobre la corba i una altra sobre la derivada, anomenades condicions inicials, expressades en general en la forma

$$\begin{cases} y(0) = y_0 \\ y'(0) = y'_0 \end{cases}$$

Aquestes condicions permeten calcular el valor dels paràmetres i determinar la solució particular corresponent. Per exemple en la família

$$y = C_1 \cos(2t) + C_2 \sin(2t), \quad C_1, C_2 \in \mathbb{R}$$

si les condicions inicials són:

$$\begin{cases} y(0) = 1 \\ y'(0) = 3 \end{cases}$$

substituint aquestes condicions en la solució general i en la derivada s'obté $C_1 = 1$, $C_2 = \frac{3}{2}$ i, per tant, la solució particular buscada és $y = \cos(2t) + \frac{3}{2} \sin(2t)$.

Càlculs amb wxMaxima:

```
(%i1) y(t):=C1*cos(2*t) + C2*sin(2*t);
      define ( Dy(t) , diff(y(t),t) ) ;
(%o1) y(t):= C1 cos(2 t)+C2 sin(2 t)
      (%o2) Dy(t):= 2 cos(2 t) C2 - 2 sin(2 t) C:
(%i3) Eq1: y(0)=1;
      Eq2: Dy(0)=3;
(%o3) C1 = 1
      (%o4) 2 C2 = 3
(%i5) C1=C1:1;      C2=C2:3/2;
(%o5) C1 = 1
      (%o6) C2 =  $\frac{3}{2}$ 
(%i7) 'y(t) = y(t) ;
(%o7) y(t) =  $\frac{3 \sin(2 t)}{2} + \cos(2 t)$ 
```

12-2.- Equacions diferencials ordinàries d'ordre 1. Sistemes.

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_12-2.wxm**.

Una equació diferencial ordinària (EDO) d'ordre 1 o de primer ordre és una equació de la forma $F(t, y, y') = 0$ en la qual t indica una variable i y una funció de t .

Resoldre o integrar una EDO és determinar la família de corbes que compleix l'equació.

S'anomena solució o integral general d'una EDO de primer ordre, una família de corbes planes de la forma $\Phi(t, y, C) = 0$ que compleix l'equació diferencial, on C és un paràmetre o constant arbitrària.

S'anomena solució o integral particular d'una EDO de primer ordre, una corba plana que forma part de la solució general, en la qual s'ha determinat el valor concret del paràmetre. Per determinar el paràmetre cal tenir una informació que s'anomena condició inicial, donada generalment mitjançant $y(t_0) = y_0$.

No és el nostre objectiu desenvolupar una teoria completa d'EDO, sinó mostrar la metodologia de resolució amb wxMaxima. En tot cas, es comentaran els principals tipus d'EDO d'ordre 1, la metodologia i sintaxi de càlcul de la solució general i d'una solució particular i es comentaran els casos en què aquest procés no es pot dur a terme fàcilment amb wxMaxima.

Els tipus d'EDO d'ordre 1 que es comentaran en aquest tema són els següents:

- De variables separades (o separables)
- Lineals
- Reduïbles a lineals (Bernoulli i Ricatti).

Hi ha altres tipus d'EDO que en el context d'un desenvolupament teòric requereixen desenvolupar la metodologia de resolució; en aquest cas és irrellevant ja que el programa les resol aplicant la mateixa metodologia.

12.2.1.- Equacions de variable separades

Una EDO de primer ordre es diu de variables separades o separables, si és de la forma

$$P(t) + Q(y)y' = 0.$$

També s'escriu sovint en la forma

$$P(t)dt + Q(y)dy = 0.$$

La solució general d'una EDO de variables separades s'obté calculant la integral indefinida de les funcions $P(t)$, $Q(y)$ i s'expressa mitjançant:

$$\int P(t)dt = -\int Q(y)dy + C$$

Exemple 12.2.1. Resoldre l'equació diferencial $2y' - y = 0$.

En primer lloc cal escriure l'equació diferencial i assignar-li una referència:

```
(%i1) EDO_vs1: 2*diff(y(t),t) - y(t) = 0;
(%o1) 2  $\left(\frac{d}{dt} y(t)\right) - y(t) = 0$ 
```


A continuació cal anar al menú “Ecuaciones” i seleccionar l'opció “Resolver EDO”, tal com es pot veure a continuació:



A la primera casella cal escriure la referència assignada a l'equació diferencial: en aquest cas és EDO_vs1. A la segona casella cal posar la funció amb indicació de la variable independent entre parèntesi: en aquest cas y(t). A la tercera casella cal indicar senzillament la variable independent: en aquest cas t. Quan s'aplica aquesta sintaxi, resulta:

```
(%i2) ode2(EDO_vs1, y(t), t);
(%o2) y(t) = %c %et/2
```


Per tant, la solució general és la família uniparamètrica $y(t) = Ce^{t/2}$, $C \in \mathbb{R}$. Si es vol resoldre un problema de valor inicial, és a dir, determinar una solució particular, cal anar al menú “Ecuaciones” i seleccionar l'opció “Problema de valor inicial (1)”, tal com es pot veure a continuació:



A la primera casella cal escriure el signe “%” si aquesta instrucció s’aplica immediatament després de l’anterior, ja que el programa interpreta que ens referim a la solució general acabada de calcular. A la segona casella cal posar el valor de la variable independent: en aquest cas $t = 0$. A la tercera casella cal indicar el valor de la funció en el punt anterior: en aquest cas $y(0) = 100$. Quan s’aplica aquesta sintaxi, resulta:

```
(%i3) ic1( % , t=0, y(0)=100);
(%o3) y(t)=100 %et/2
```

També es pot calcular la solució particular seleccionant la solució en l’output de wxMaxima corresponent i anant al quadre de diàleg anterior, on s’observarà que la solució s’enganxa a la primera casella. El resultat és




En wxMaxima s’obté:

```
(%i4) ic1( y(t)=%c*%e^(t/2) , t=0, y(0)=100);
(%o4) y(t)=100 %et/2
```

Si es vol es pot obtenir una representació gràfica d’aquesta solució particular:

```
(%i5) y(t):=100*%e^(t/2)$
wxplot2d([y(t)], [t, 0, 5], [ylabel, ""], [style, [lines, 2]],
[gnuplot_preamble, "set size ratio 1; set grid;"], [nticks, 100])$
```

(%t6)



Exemple 12.2.2. Resoldre l'equació diferencial $y' - 2(100 - y) = 0$.

Vegem la resolució amb wxMaxima d'acord amb el procediment descrit a l'exemple anterior:


```
(%i1) EDO_vs2:diff(y(x),x)=2*(100-y(x));
(%o1)  $\frac{d}{dx} y(x) = 2(100 - y(x))$ 

(%i2) ode2(EDO_vs2, y(x), x);
(%o2)  $y(x) = \%e^{-2x}(100 \%e^{2x} + \%c)$ 
(%i3) expand(%);
(%o3)  $y(x) = \%c \%e^{-2x} + 100$ 

(%i4) ic1(%, x=0, y(0)=50);
(%o4)  $y(x) = \%e^{-2x}(100 \%e^{2x} - 50)$ 
(%i5) expand(%);
(%o5)  $y(x) = 100 - 50 \%e^{-2x}$ 

(%i6) y(x):=100-50*%e^(-2*x)$
wxplot2d([y(x)], [x,0, 3], [ylabel, ""], [style, [lines, 2]],
[gnuplot_preamble, "set size ratio 1; set grid;"], [nticks, 100])$
```

(%t7)



12.2.2.- Equacions diferencials lineals

S'anomenen EDO lineals de primer ordre les equacions de la forma:

$$y' + G(t)y = F(t)$$

on F, G són funcions reals contínues i derivables en un cert obert $\Omega \subset \mathbb{R}$. Si $F(t) = 0$ l'equació es diu homogènia i si $F(t) \neq 0$ l'equació es diu completa.

Com és sabut, la solució general de l'equació completa s'obté sumant la solució general de l'equació homogènia amb una solució particular de l'equació completa, és a dir:

$$y(t) = y_H(t) + y_p(t) = K e^{-\int G(t)dt} + e^{-\int G(t)dt} \int F(t) e^{\int G(t)dt} dt, \quad K \in \mathbb{R}$$

Exemple 12.2.3.- Considerem l'equació diferencial $t^2 - y = ty'$; es tracta d'una equació lineal ja que es pot escriure:

$$y' + \frac{1}{t}y = t; \quad G(t) = \frac{1}{t}, \quad F(t) = t$$

Vegem la resolució amb wxMaxima:

```
(%i1) EDO_lin1: t^2 - y(t) = t * diff(y(t), t);
(%o1) t^2 - y(t) = t * (d/dt y(t))
```

```
(%i2) ode2(EDO_lin1, y(t), t);
```

```
(%o2) y(t) = (t^3/3 + %c)/t
```

```
(%i3) expand(%);
```

```
(%o3) y(t) = t^2/3 + %c/t
```

Per tant, la solució general de l'equació considerada és:

$$y(t) = C \frac{1}{t} + \frac{t^2}{3}, \quad C \in \mathbb{R}$$

Observeu que la solució general es dona directament i es pot interpretar cada un dels termes d'acord amb l'estructura de la solució general comentada abans.

Si es vol calcular una solució particular, per exemple, la que compleix les condicions inicials $t = 1, y(1) = 5$, s'obté:

```
(%i4) ic1(y(t)=t^2/3+%c/t, t=1, y(1)=5);
```

```
(%o4) y(t)= $\frac{t^3+14}{3t}$ 
```


```
(%i5) expand(%);
```

```
(%o5) y(t)= $\frac{t^2}{3} + \frac{14}{3t}$ 
```

Finalment es pot fer una representació gràfica d'aquesta solució particular:

```
(%i6) y(t):=t^2/3+14/(3*t)$  
wxplot2d([y(t)], [t, 1, 10], [ylabel, ""], [style, [lines, 2]],  
[gnuplot_preamble, "set size ratio 1; set grid;"], [nticks, 100])$
```

```
(%t7)
```



12.2.3.- Equacions de Bernoulli

S'anomenen equacions de Bernoulli les equacions que tenen la forma general

$$y' + G(t)y = F(t)y^n, \quad n \in \mathbb{N}$$

Observeu que si $n = 0, n = 1$ es tracta aleshores d'una equació lineal. Per resoldre les equacions de Bernoulli es duu a terme el canvi de funció definit per:

$$u = y^{1-n} \quad (n \geq 2)$$

Aleshores es compleix:

$$u' = (1-n)y^{-n}y' \Rightarrow y' = \frac{1}{1-n}y^n u'$$

Substituint a l'equació inicial resulta:

$$u' + (1-n)G(t)u = (1-n)F(t) \quad (n \geq 2)$$

que és una equació lineal i que ja se sap com resoldre.

Exemple 12.2.4. Es considera l'equació diferencial:

$$y' + 2ty + ty^4 = 0$$

Es tracta d'una equació de Bernoulli:

$$y' + 2ty = -ty^4; \quad F(t) = -t, \quad G(t) = 2t$$

Vegem la resolució amb wxMaxima:

```
(%i1) EDO_ber1: diff(y(t), t) + 2*t*y(t) + t*y(t)^4=0;
```

```
(%o1)  $\frac{d}{dt} y(t) + t y(t)^4 + 2 t y(t) = 0$ 
```

```
(%i2) ode2(EDO_ber1, y(t), t);
```

```
(%o2)  $\frac{\log(y(t)^3 + 2) - \log(y(t)^3)}{6} = \frac{t^2}{2} + \%c$ 
```

```
(%i3) logcontract(%);
```

```
(%o3)  $-\frac{\log\left(\frac{y(t)^3}{y(t)^3 + 2}\right)}{6} = \frac{t^2}{2} + \%c$ 
```

Ara expressarem la solució en la forma més simple possible:

```
(%i4) log((y(t)^3 + 2)/(y(t)^3))=3*t^2+C1;
```

```
(%o4)  $\log\left(\frac{y(t)^3 + 2}{y(t)^3}\right) = C1 + 3 t^2$ 
```

```
(%i5) ((y(t)^3+2)/y(t)^3)=exp(C1+3*t^2);
```

```
(%o5)  $\frac{y(t)^3 + 2}{y(t)^3} = \%e^{C1 + 3 t^2}$ 
```

```
(%i6) eq1: (y(t)^3+2)/y(t)^3=%e^(C1+3*t^2)$
solve(eq1, y(t));
```

```
(%o7) [y(t)= $\frac{2^{1/3}\sqrt[3]{3}\%i-2^{1/3}}{2(\%e^{C1+3t^2}-1)^{1/3}}$ , y(t)= $-\frac{2^{1/3}\sqrt[3]{3}\%i+2^{1/3}}{2(\%e^{C1+3t^2}-1)^{1/3}}$ , y(t)= $\frac{2^{1/3}}{(\%e^{C1+3t^2}-1)^{1/3}}$ ]
```

Per tant, la solució buscada és:

```
(%i8) y(t):=2^(1/3)/(%e^(C1+3*t^2)-1)^(1/3);
```

```
(%o8) y(t):= $\frac{2^{1/3}}{(\%e^{C1+3t^2}-1)^{1/3}}$ 
```

Calculem ara una solució particular, aplicant la condició inicial $y(0) = 3$. Es tindrà:

```
(%i9) y(0);
```

```
(%o9)  $\frac{2^{1/3}}{(\%e^{C1}-1)^{1/3}}$ 
```

```
(%i10) eq2: y(0)=1;
solve(eq2, C1);
```

```
(%o10)  $\frac{2^{1/3}}{(\%e^{C1}-1)^{1/3}}=1$ 
```

```
(%o11) [C1=log(3)]
```

Per tant, la solució particular és:


```
(%i12) yp(t):=2^(1/3)/(%e^(log(3)+3*t^2)-1)^(1/3);
```

```
(%o12) yp(t):= $\frac{2^{1/3}}{(\%e^{\log(3)+3t^2}-1)^{1/3}}$ 
```

Representació gràfica d'aquesta solució particular:

```
(%i13) wxplot2d([yp(t)], [t, 0, 3], [ylabel, ""], [style, [lines, 2]],  
[gnuplot_preamble, "set size ratio 1; set grid;"], [nticks, 100])$
```

(%t13)



12.2.4.- Equacions de Ricatti

S'anomenen equacions de Ricatti les equacions que tenen la forma general

$$y' = F_1(t) + F_2(t)y + F_3(t)y^2$$

en la qual, les funcions $F_1(t), F_2(t), F_3(t)$ són contínues i derivables en un cert interval obert de la recta real. Per resoldre una equació de Ricatti es duu a terme el canvi de funció definit per:

$$y = f_1 + \frac{1}{u}; \quad y' = f_1' - \frac{u'}{u^2}; \quad u = \frac{1}{y - f_1}$$

on la funció $f_1(t)$ és una solució particular de l'equació de Ricatti i que cal conèixer prèviament. Aleshores substituint a l'equació de Ricatti resulta:

$$u' + (F_2(t) + 2F_3(t)f_1(t))u = -F_3(t)$$

que és una equació lineal i que ja se sap com resoldre.

Exemple 12.2.5. Es considera l'equació diferencial de Ricatti:

$$t(t-1)y' - (1-2t)y + y^2 - 2t = 0$$

Vegem com es pot resoldre amb wxMaxima:

```
(%i1) EDO_ric1: t*(t-1)*diff(y(t), t) - (1-2*t)*y(t) + y(t)^2 = 2*t;
(%o1) (t-1)t\left(\frac{d}{dt}y(t)\right)+y(t)^2-(1-2t)y(t)=2t

(%i2) ode2(EDO_ric1, y(t), t);
(%o2) false
```

Com podem observar, el programa no la pot resoldre amb aquesta instrucció. Aplicarem la metodologia de resolució exposada per aquest tipus d'equacions. En primer lloc observem que la funció $f_1(t) = 1$ és una solució particular de l'equació, amb la qual cosa farem el canvi de funció definit per:

```
(%i3) y(t):=1+1/u(t);
(%o3) y(t):=1+\frac{1}{u(t)}
```

Ara calculem la derivada i substituïm a l'equació donada:

```
(%i4) diff(y(t), t);
(%o4) -\frac{\frac{d}{dt}u(t)}{u(t)^2}

(%i5) EDO_ric2: t*(t-1)*(-'diff(u(t),t,1)/u(t)^2) - (1-2*t)*y(t) + y(t)^2 = 2*t;
(%o5) -\frac{(t-1)t\left(\frac{d}{dt}u(t)\right)}{u(t)^2}-(1-2t)\left(\frac{1}{u(t)}+1\right)+\left(\frac{1}{u(t)}+1\right)^2=2t

(%i6) ratsimp(%);
(%o6) -\frac{(t^2-t)\left(\frac{d}{dt}u(t)\right)-2tu(t)^2+(-2t-1)u(t)-1}{u(t)^2}=2t
```

Simplifiquem aquesta equació:

```
(%i7) EDO_ric3: (t^2-t)*('diff(u(t),t,1)) + (-2*t-1)*u(t) - 1 = 0;
(%o7) (t^2-t)\left(\frac{d}{dt}u(t)\right)+(-2t-1)u(t)-1=0
```

Calculem la solució general d'aquesta equació:

```
(%i8) ode2(EDO_ric3, u(t), t);
```

$$(\%o8) u(t) = \frac{(t-1)^3 \left(\frac{1}{3t^3 - 9t^2 + 9t - 3} \right)}{t}$$

Simplificant novament, s'obté:

```
(%i10) u(t):=C*(t-1)^3/t - 1/(3*t);
```

$$(\%o10) u(t) := \frac{C(t-1)^3}{t} - \frac{1}{3t}$$

Per tant, la solució general de l'equació considerada és:

```
(%i12) 'y(t)=y(t);
```

$$(\%o12) y(t) = \frac{1}{\frac{(t-1)^3 C}{t} - \frac{1}{3t}} + 1$$

Podem simplificar aquesta expressió:

```
(%i13) ratsimp(%);
```

$$(\%o13) y(t) = \frac{(3t^3 - 9t^2 + 9t - 3)C + 3t - 1}{(3t^3 - 9t^2 + 9t - 3)C - 1}$$

```
(%i14) y(t):=(3*C*(t-1)^3 + 3*t - 1) / (3*C*(t-1)^3 - 1);
```

$$(\%o14) y(t) := \frac{3C(t-1)^3 + 3t - 1}{3C(t-1)^3 - 1}$$

Ara calculem la solució particular tal que $y(2) = 2$:

```
(%i15) eq5: y(2)=2;
solve(eq5, C);
```

$$(\%o15) \frac{3C+5}{3C-1} = 2$$

$$(\%o16) [C = \frac{7}{3}]$$

Per tant:


```
(%i17) yp(t):=(3*(7/3)*(t-1)^3+3*t-1)/(3*(7/3)*(t-1)^3-1);
```

$$(\%o17) yp(t) := \frac{3\frac{7}{3}(t-1)^3 + 3t - 1}{3\frac{7}{3}(t-1)^3 - 1}$$

Representació gràfica:

```
(%i18) wxplot2d([yp(t)], [t, 2, 5], [ylabel, ""] , [style, [lines, 2]] ,
[gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 100] )$
```

(%t18)



12.2.5.- Sistemes d'EDO de primer ordre

Un sistema d'EDO d'ordre 1 o d'EDO de primer ordre és un sistema d'equacions diferencials ordinàries de la forma:

$$\begin{cases} x_1' = f_1(t, x_1, x_2, \dots, x_n) \\ x_2' = f_2(t, x_1, x_2, \dots, x_n) \\ \dots \\ x_n' = f_n(t, x_1, x_2, \dots, x_n) \end{cases}$$

Un cas particular molt important és el dels sistemes d'EDO lineals d'ordre 1, que són sistemes d'EDO en el qual les funcions del segon membre de cada equació són combinacions lineals de les variables dependents; aquestes combinacions poden ser de coeficients constants o no. Així, un sistema d'EDO lineals de primer ordre és:

$$\begin{cases} x_1' = a_1^1(t)x_1 + a_2^1(t)x_2 + \dots + a_n^1(t)x_n \\ x_2' = a_1^2(t)x_1 + a_2^2(t)x_2 + \dots + a_n^2(t)x_n \\ \dots \\ x_n' = a_1^n(t)x_1 + a_2^n(t)x_2 + \dots + a_n^n(t)x_n \end{cases}$$

A continuació il·lustrarem la metodologia de resolució de sistemes d'EDO de primer ordre amb wxMaxima.

Exemple 12.2.6.- Es tracta de resoldre el sistema d'EDO de primer ordre:

$$\begin{cases} x' + y' - x = 2t + 1 \\ 2x' + 2y' + x = t \end{cases}$$

En primer lloc es tracta d'escriure cada una de les EDO del sistema i assignar en cada cas una referència:

```
(%i1) EDO_1a: 'diff(x(t), t) - x(t) + 'diff(y(t), t) = 2*t + 1;
      EDO_1b: 2*'diff(x(t), t) + x(t) + 2*'diff(y(t), t) = t ;
(%o1)  $\frac{d}{dt}y(t) + \frac{d}{dt}x(t) - x(t) = 2t + 1$ 
      (%o2)  $2\left(\frac{d}{dt}y(t)\right) + 2\left(\frac{d}{dt}x(t)\right) + x(t) = t$ 
```

A continuació cal escriure la instrucció que duu a terme el càlcul de la solució general del sistema, que en aquest cas és la instrucció “desolve”:

```
(%i3) desolve( [EDO_1a, EDO_1b] , [x(t), y(t)] );
(%o3)  $[x(t) = -t - \frac{2}{3}, y(t) = \frac{t^2}{2} + \frac{4t}{3} + \frac{3y(0) + 3x(0) + 2}{3}]$ 
```

Si es vol calcular una solució particular, cal veure les condicions inicials necessàries per fer-ho, que depenen del nombre de paràmetres de la solució general; en aquest cas n'hi ha dos i la sintaxi amb wxMaxima és:

```
(%i4) atvalue(x(t), t=0, -2/3);
      atvalue(y(t), t=0, 1);
(%o4)  $-\frac{2}{3}$ 
      (%o5) 1
```


Finalment es demana al programa el càlcul de la solució particular que correspon a aquestes condicions inicials:

```
(%i6) desolve( [EDO_1a, EDO_1b] , [x(t), y(t)] );
(%o6)  $[x(t) = -t - \frac{2}{3}, y(t) = \frac{t^2}{2} + \frac{4t}{3} + 1]$ 
```

Com en exemples anteriors, il·lustrem aquesta solució particular amb una representació gràfica, que té com a pas previ la definició de les solucions particulars com funcions:

```
(%i7) x(t):=-t-2/3$ y(t):=t^2/2+(4*t)/3+1$
wxplot2d([x(t), y(t)], [t, 0, 10], [legend, "x(t)", "y(t)"] , [style, [lines, 2]] ,
[gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 100] )$
```

(%t9)



Exemple 12.2.7.- Es tracta de resoldre el sistema d'EDO de primer ordre:

$$\begin{cases} x' + 2y' - 3x + 4y = 2\sin(t) \\ 2x' + y' + 2x - y = t \end{cases}$$

Escriurem cada una de les EDO del sistema i assignarem en cada cas una referència:

```
(%i1) EDO_2a: 'diff(x(t), t) - 3*x(t) + 2*'diff(y(t), t) + 4*y(t)= 2*sin(t) ;
EDO_2b: 2*'diff(x(t), t) + 2*x(t) + 'diff(y(t), t) - y(t) = t ;
```

```
(%o1) 2  $\left(\frac{d}{dt} y(t)\right) + \frac{d}{dt} x(t) + 4 y(t) - 3 x(t) = 2 \sin(t)$ 
```

```
(%o2)  $\frac{d}{dt} y(t) + 2 \left(\frac{d}{dt} x(t)\right) - y(t) + 2 x(t) = t$ 
```

A continuació calcularem la solució general del sistema:

```
(%i3) solve( [EDO_2a, EDO_2b] , [x(t), y(t)] );
```

```
(%o3) [x(t)=- $\frac{7 \sin(t)}{65}$ - $\frac{9 \cos(t)}{65}$ + $\frac{(45 y(0)+60 x(0)+243) \%e^{-\frac{t}{3}}}{105}$ - $\frac{(975 y(0)-975 x(0)+36) \%e^{-5 t}}{2275}$ + $\frac{4 t}{5}$ - $\frac{54}{25}$ , y(t)=- $\frac{18 \sin(t)}{65}$ - $\frac{14 \cos(t)}{65}$ + $\frac{(45 y(0)+60 x(0)+243) \%e^{-\frac{t}{3}}}{105}$ + $\frac{(1300 y(0)-1300 x(0)+48) \%e^{-5 t}}{2275}$ + $\frac{3 t}{5}$ - $\frac{53}{25}$ ]
```


Per calcular una solució particular, cal introduir ara les condicions inicials necessàries per fer-ho:

```
(%i4) atvalue(x(t), t=0, 2);
      atvalue(y(t), t=0, 3);
(%o4) 2
      (%o5) 3
```

Finalment calcularem la solució particular que correspon a aquestes condicions inicials:


```
(%i6) desolve( [EDO_2a, EDO_2b] , [x(t), y(t)] );
```

$$(\%o6) \left[x(t) = -\frac{7 \sin(t)}{65} - \frac{9 \cos(t)}{65} + \frac{166}{35} e^{-\frac{t}{3}} - \frac{1011}{2275} e^{-5t} + \frac{4t}{5} - \frac{54}{25}, y(t) = \frac{18 \sin(t)}{65} - \frac{14 \cos(t)}{65} + \frac{166}{35} e^{-\frac{t}{3}} + \frac{1348}{2275} e^{-5t} + \frac{3t}{5} - \frac{53}{25} \right]$$

Com en exemples anteriors, il·lustrem aquesta solució particular amb una representació gràfica, que té com a pas previ la definició de les solucions particulars com funcions:

```
(%i9) wxplot2d([x(t), y(t)], [t, 0, 10], [legend, "x(t)", "y(t)"] , [style, [lines, 2]] ,
[gnuplot_preamble, "set size ratio 1; set grid;"] , [nticks, 100] )$
```

(%t9)



12-3.- Equacions diferencials ordinàries d'ordre 2

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_12-3.wxm**.

Una equació diferencial ordinària (EDO) d'ordre 2 o de segon ordre és una equació de la forma $F(t, y, y', y'') = 0$ en la qual t indica una variable i y una funció de t .

Resoldre o integrar una EDO és determinar la família de corbes que compleix l'equació.

S'anomena solució general o integral general d'una EDO de primer ordre, una família de corbes planes de la forma $\Phi(t, y, C_1, C_2) = 0$ que compleix l'equació diferencial, on C_1, C_2 són dos paràmetres o constants arbitràries.

S'anomena solució particular o integral particular d'una EDO d'ordre 2, una corba plana que forma part de la solució general, en la qual s'han determinat els valors concrets dels paràmetres. Per determinar els paràmetres cal tenir una informació que s'anomena condicions inicials, donades generalment mitjançant

$$\begin{cases} y(t_0) = y_0 \\ y'(t_0) = y'_0 \end{cases}$$

Com en l'apartat anterior, no és el nostre objectiu desenvolupar una teoria completa d'EDO, sinó mostrar la metodologia de resolució amb wxMaxima. En aquest cas, es veurà la metodologia per resoldre una EDO d'ordre 2 i es comentarà un tipus d'EDO d'ordre 2 d'importància especial: les EDO lineals de coeficients constants.

Exemple 12.3.1.- Es vol resoldre l'EDO d'ordre 2

$$y'' - y(y')^3 = 0$$

Vegem com es pot resoldre amb wxMaxima. En primer lloc s'introdueix l'equació i se li dóna una referència:

```
(%i1) EDO_1: 'diff(y, t, 2) - y*'diff(y, t)^3 = 0;
```

$$(\%o1) \frac{d^2}{dt^2} y - y \left(\frac{d}{dt} y \right)^3 = 0$$

A continuació s'aplica la instrucció per a la resolució d'equacions diferencials:

```
(%i2) ode2(EDO_1, y, t);
```

$$(\%o2) -\frac{y^3 + 6 \%k1 y}{6} = t + \%k2$$

Aquesta és la solució general, que conté dos paràmetres. Per calcular una solució particular aplicarem unes condicions inicials:

```
(%i3) ic2(% , t=0, y=0, 'diff(y,t)=2);
```

```
(%o3) 
$$-\frac{y^3 - 3y(y^2 + 1)}{6} = t$$

```


```
(%i4) ratsimp(%);
```

```
(%o4) 
$$\frac{2y^3 + 3y}{6} = t$$

```

Ara podem obtenir una representació gràfica; en aquest cas podem fer-ho amb la corba donada en forma implícita:

```
(%i5) load(draw)$
      wxdraw2d( grid = true, line_type = solid, line_width = 2, color = blue,
      implicit( (2*y^3+3*y)/6=t, t, -2.0, 2.0, y, -2,2)
      );
```



```
(%o6) [gr2d(implicit)]
```

Exemple 12.3.2.- Es vol resoldre l'EDO d'ordre 2

$$t^2 y'' - t y' + 4y = 0$$

Vegem com es pot resoldre amb wxMaxima. En primer lloc s'introdueix l'equació i se li dona una referència:

```
(%i1) EDO_2: t^2*'diff(y, t, 2) - t*'diff(y, t) + 4*y = 0;
(%o1)  $t^2 \left( \frac{d^2}{dt^2} y \right) - t \left( \frac{d}{dt} y \right) + 4y = 0$ 
```

A continuació s'aplica la instrucció per a la resolució d'equacions diferencials:


```
(%i2) ode2(EDO_2, y, t);
(%o2)  $y = t (\%k1 \sin(\sqrt{3} \log(t)) + \%k2 \cos(\sqrt{3} \log(t)))$ 
```

Aquesta és la solució general, que conté dos paràmetres. Per calcular una solució particular aplicarem unes condicions inicials:

```
(%i3) ic2(%o2, t=1, y=1, 'diff(y,t)=2);
(%o3)  $y = t \left( \frac{\sin(\sqrt{3} \log(t))}{\sqrt{3}} + \cos(\sqrt{3} \log(t)) \right)$ 
(%i4) expand(%o3);
(%o4)  $y = \frac{t \sin(\sqrt{3} \log(t))}{\sqrt{3}} + t \cos(\sqrt{3} \log(t))$ 
```

Ara podem obtenir una representació gràfica; en aquest cas podem fer-ho amb la corba donada en forma implícita:

```
(%i5) load(draw)$
      wxdraw2d( grid = true, line_type = solid, line_width = 2, color = blue,
      implicit( y=(t*sin(sqrt(3)*log(t)))/sqrt(3)+t*cos(sqrt(3)*log(t)) , t, 1, 10, y, -12,2)
      );
rat: replaced 1.732050807568877 by 5042/2911 = 1.732050841635177
rat: replaced 1.732050807568877 by 5042/2911 = 1.732050841635177
rat: replaced 1.732050807568877 by 5042/2911 = 1.732050841635177
```



Si es prefereix fer una gràfica amb la solució donada en forma explícita primer cal definir-la:

```
(%i7) y(t):=(t*sin(sqrt(3)*log(t))/sqrt(3)+t*cos(sqrt(3)*log(t)));
```


```
(%o7) 
$$y(t) := \frac{t \sin(\sqrt{3} \log(t))}{\sqrt{3}} + t \cos(\sqrt{3} \log(t))$$

```

I ara ja es pot fer la representació gràfica:

```
(%i8) wxplot2d([y(t)], [t, 1, 10], [ylabel, ""], [style, [lines, 2]],  
[gnuplot_preamble, "set size ratio 1; set grid;"], [nticks, 100])$
```

```
(%t8)
```



Exemple 12.3.3.- Es vol resoldre l'EDO d'ordre 2

$$y'' - \frac{3}{t}y' + \frac{3y}{t^2} = 2t - 1$$

Introduïm l'equació i li donem una referència:

```
(%i1) EDO_3: 'diff(y, t, 2) - (3/t) * 'diff(y, t) + (3/t^2)*y = 2*t - 1;
```

```
(%o1) 
$$\frac{d^2}{dt^2}y - \frac{3 \left( \frac{d}{dt}y \right)}{t} + \frac{3y}{t^2} = 2t - 1$$

```

A continuació s'aplica la instrucció per a la resolució d'equacions diferencials:

```
(%i2) ode2(EDO_3, y, t);
```

```
(%o2) 
$$y = \frac{2 t^3 \log(t) - t^3 + 2 t^2}{2} + \%k1 t^3 + \%k2 t$$

```

```
(%i3) expand(%);
```

```
(%o3) y = t^3 log(t) + %k1 t^3 - t^3/2 + t^2 + %k2 t
```

Aquesta és la solució general, que conté dos paràmetres. Per calcular una solució particular aplicarem unes condicions inicials:

```
(%i4) ic2(% , t=1, y=1, 'diff(y,t)=2);
```


```
(%o4) y = t^3 log(t) - t^3/2 + t^2 + t/2
```

```
(%i5) expand(%);
```

```
(%o5) y = t^3 log(t) - t^3/2 + t^2 + t/2
```

Ara podem obtenir una representació gràfica; en aquest cas podem fer-ho amb la corba donada en forma implícita:

```
(%i6) load(draw)$
      wxdraw2d( grid = true, line_type = solid, line_width = 2, color = blue,
      implicit( y=t^3*log(t)-t^3/2+t^2+t/2 , t, 1, 2, y, 0,8)
      );
```



```
(%o7) [gr2d(implicit)]
```

Es tracta a continuació d'estudiar un tipus concret d'EDO d'ordre 2: les **equacions diferencials lineals de segon ordre i de coeficients constants**.

S'anomenen equacions diferencials lineals de segon ordre i coeficients constants les de la forma:

$$a_2 y'' + a_1 y' + a_0 y = F(t), \quad a_2, a_1, a_0 \in \mathbb{R}, a_2 \neq 0$$

Si $F(t)=0$ l'equació s'anomena homogènia; si $F(t) \neq 0$ aleshores l'equació s'anomena completa.

Recordem que si D indica l'operador derivació, és a dir, l'aplicació lineal que assigna a una funció derivable la seva funció derivada, aleshores l'equació es pot escriure:

$$a_2 D^2 y + a_1 D y + a_0 y = F(t), \quad a_2, a_1, a_0 \in \mathbb{R}, a_2 \neq 0$$

i també:

$$(a_2 D^2 + a_1 D + a_0 I) y = L(D)y = F(t), \quad a_2, a_1, a_0 \in \mathbb{R}, a_2 \neq 0$$

La metodologia de resolució de l'equació diferencial lineal de segon ordre i coeficients constants es fonamenta en el resultat que s'enuncia a continuació.

Teorema.- Si $y_H(t)$ designa la solució general de l'equació homogènia $L(D)y = 0$ i $y_P(t)$ és una solució particular de l'equació completa $L(D)y = F(t)$, aleshores la solució general de l'equació completa és $y(t) = y_H(t) + y_P(t)$.

Per tant, en virtut del teorema anterior, per resoldre una equació diferencial lineal de segon ordre i coeficients constants cal calcular d'una banda la solució general de l'equació homogènia associada i de l'altra calcular una solució particular de l'equació completa. Això comporta desenvolupar una metodologia per a la resolució de l'equació homogènia i una altra per calcular una solució particular de l'equació completa; el mètode aplicat clàssicament es coneix com mètode de variació dels paràmetres. Amb wxMaxima no cal aplicar aquest procediment, ja que s'obté directament la solució.

Exemple 12.3.4.- Es tracta de calcular la solució general de l'equació diferencial

$$y'' + 4y = 12t.$$

Introduïm l'equació i li donem una referència:

```
(%i1) EDO_4: diff(y(t), t, 2) + 4*y(t) = 12*t;
```

$$(\%o1) \frac{d^2}{dt^2} y(t) + 4 y(t) = 12 t$$

A continuació s'aplica la instrucció per a la resolució d'equacions diferencials:

```
(%i2) ode2(EDO_4, y(t), t);
```

$$(\%o2) y(t) = \%k1 \sin(2 t) + \%k2 \cos(2 t) + 3 t$$

Aquesta és la solució general, que conté dos paràmetres. Observem que els primers dos termes corresponen a la solució general de l'equació homogènia associada; en efecte:

```
(%i3) EDO_5: diff(y(t), t, 2) + 4*y(t) = 0;
```

```
(%o3)  $\frac{d^2}{dt^2} y(t) + 4 y(t) = 0$ 
```

```
(%i4) ode2(EDO_5, y(t), t);
```

```
(%o4)  $y(t) = \%k1 \sin(2 t) + \%k2 \cos(2 t)$ 
```


Per calcular una solució particular aplicarem unes condicions inicials:

```
(%i5) ic2( y=%k1*sin(2*t)+%k2*cos(2*t)+3*t, t=0, y=3, diff(y,t)=2);
```

```
(%o5)  $y = -\frac{\sin(2 t)}{2} + 3 \cos(2 t) + 3 t$ 
```

Ara podem obtenir una representació gràfica; en aquest cas podem fer-ho amb la corba donada en forma implícita:

```
(%i6) load(draw)$
wxdraw2d( grid = true, line_type = solid, line_width = 2, color = blue,
implicit( y=-sin(2*t)/2+3*cos(2*t)+3*t, t, 0, 10, y, 0,40)
);
```



```
(%o7) [gr2d(implicit)]
```


12-4.- Introducció als mètodes numèrics de resolució d'equacions diferencials ordinàries i sistemes

Els continguts d'aquest apartat es desenvolupen al fitxer **Tema_12-4.wxm**.

Malgrat el gran nombre de mètodes que permeten el càlcul de la solució d'una equació diferencial ordinària o d'un sistema d'equacions diferencials ordinàries, en molts casos el càlcul de la solució analítica pot resultar inabordable o, fins i tot, impossible per la no existència de solució analítica. En aquests casos es pot recórrer a mètodes aproximats, els més importants dels quals són els mètodes numèrics. La potència de càlcul de programes com wxMaxima proporciona una eina molt útil per resoldre aquestes situacions. En aquest apartat es descriu la metodologia i la sintaxi per resoldre EDO mitjançant mètodes numèrics, en particular el mètode programat en wxMaxima que es coneix com mètode de Runge-Kutta.

La metodologia consisteix essencialment en dividir l'interval $[a, b]$ en el qual es vol calcular la solució mitjançant un conjunt de punts $P = \{t_0, t_1, \dots, t_n\}$ anomenat partició en el qual $h = t_{k+1} - t_k$ és constant i fixat per l'usuari. Això dona lloc a un conjunt d'interval $[t_k, t_{k+1}]$ en l'extrem dels quals s'aproxima el valor de la funció solució particular de l'equació diferencial. No entrarem en detalls sobre la precisió del mètode i el càlcul de la fita l'error comés; en aquest Curs es tracta senzillament d'il·lustrar l'aplicació del mètode de Runge-Kutta de quart ordre implementat en wxMaxima, cosa que farem mitjançant els exemples que segueixen.

Exemple 12.4.1.- Es tracta de calcular la solució particular de l'equació diferencial

$$t(t-1)y' - (1-2t)y + y^2 = 2t.$$

amb la condició inicial $y(0) = 2$. Observeu que es tracta de l'equació de Riccati que s'ha estudiat a l'Exemple 12.2.5. Introduïm l'equació i li donem una referència:

```
(%i1) EDO: t*(t-1)*diff(y(t), t) - (1-2*t)*y(t) + y(t)^2 = 2*t;  
(%o1) (t-1)t  $\left(\frac{d}{dt}y(t)\right) + y(t)^2 - (1-2t)y(t) = 2t$ 
```

Com es recordarà, wxMaxima no pot calcular la solució general d'aquesta equació diferencial amb la instrucció a aquest efecte, tot i que sabem que aquesta solució analítica existeix. Així, recordem que la resposta de wxMaxima és:

```
(%i2) ode2(EDO, y(t), t);  
(%o2) false
```

Ens proposem doncs calcular la solució particular de l'EDO amb el mètode de Runge-Kutta, per a la qual cosa primer en primer lloc obtindrem de forma explícita la derivada de la funció:


```
(%i3) EDO_1b : (t-1)*t*Dy(t)+y(t)^2-(1-2*t)*y(t)=2*t;
(%o3) y(t)^2+(2 t-1) y(t)+(t-1) t Dy(t)= 2 t
(%i4) solve( [EDO_1b] , [Dy(t)] );
(%o4) [Dy(t)=- $\frac{y(t)^2+(2 t-1) y(t)-2 t}{t^2-t}$ ]
```

Vegem ara la sintaxi del mètode. Abans limitarem el nombre de xifres decimals de les aproximacions numèriques de la solució:

```
(%i5) fpprintprec : 5$
(%i6) soluc1: rk( [-(y^2+(2*t-1)*y-2*t)/(t^2-t)] , [y], [2], [t, 2, 5, 0.1] );
(%o6) [[2, 2], [2.1, 1.7575], [2.2, 1.5948], [2.3, 1.4799], [2.4, 1.3954], [2.5, 1.3315], [2.6, 1.2819], [2.7, 1.2426], [2.8, 1.2109], [2.9, 1.1851], [3.0, 1.1636], [3.1, 1.1457], [3.2, 1.1306], [3.3, 1.1176], [3.4, 1.1065], [3.5, 1.0969], [3.6, 1.0885], [3.7, 1.0812], [3.8, 1.0747], [3.9, 1.0689], [4.0, 1.0638], [4.1, 1.0593], [4.2, 1.055], [4.3, 1.0515], [4.4, 1.0482], [4.5, 1.0451], [4.6, 1.0424], [4.7, 1.0399], [4.8, 1.0376], [4.9, 1.0355], [5.0, 1.0336]]
```


La solució es dona en forma de llista amb els parells de valors que corresponen a la variable independent i la variable dependent. Precisament en la instrucció anterior s'assigna un nom a aquesta llista i això permet ara obtenir-ne una representació gràfica:

```
(%i6) wxplot2d( [ [discrete, soluc1] ], [style, [points, 1, 1, 2] ] );
```



Si es vol es pot canviar el format de la representació gràfica:

```
(%i7) wxplot2d( [ [discrete, soluc1] ], [style, [lines, 2, 1] ] );
```



Exemple 12.4.2.- Es tracta de calcular la solució particular de l'equació diferencial

$$(100 - t^6)y' - \cos(t) = 0.$$

amb la condició inicial $y(0) = 0.01$.

Introduïm l'equació i li donem una referència:

```
(%i1) EDO: (100 - t^6)*diff(y(t), t) - cos(t) = 0;
```

```
(%o1) (100 - t^6)  $\left(\frac{d}{dt} y(t)\right) - \cos(t) = 0$ 
```

Per obtenir la solució establim ara una precisió numèrica dels resultats de wxMaxima:


```
(%i2) fpprintprec : 5 $
```

Ara calcularem la solució particular amb el mètode de Runge-Kutta, per a la qual cosa cal tenir de forma explícita la derivada de la funció, que en aquest cas és immediata. Vegem el resultat:

```
(%i4) soluc1: rk( [ cos(t)/(100 - t^6) ], [y], [0.01], [t, 0, 1, 0.05] );
(%o4) [[0,0.01],[0.05,0.0105],[0.1,0.011],[0.15,0.0115],[0.2,0.012],[0.25,0.0125],[0.3,0.013],[0.35,0.0134],[0.4,0.0139],[0.45,0.0143],[0.5,0.0148],[0.55,0.0152],[0.6,0.0156],[0.65,0.0161],[0.7,0.0166],[0.75,0.0168],[0.8,0.0172],[0.85,0.0175],[0.9,0.0178],[0.95,0.0181],[1.0,0.0184]]
```

La solució es dona en forma de llista amb els parells de valors que corresponen a la variable independent i la variable dependent. Ara podem obtenir una representació gràfica:

```
(%i5) wxplot2d( [ [discrete, soluc1] ] , [style, [points, 1, 1, 2] ] );
```



Exemple 12.4.3.- Es tracta de calcular la solució particular del sistema d'equacions diferencials:

$$\begin{cases} x_1' = -0.2x_1 + 0.3x_2 \\ x_2' = 0.1x_1 - 0.4x_2 \end{cases}$$

amb les condicions inicials $x_1(0) = 1, x_2(0) = 3$.

Introduïm les equacions del sistema:

```
(%i4) edo1: 'diff(x1, t)= -0.2*x1 + 0.3*x2;  
edo2: 'diff(x2, t)= 0.1*x1 - 0.4*x2;
```

```
(%o4)  $\frac{d}{dt}x_1 = 0.3x_2 - 0.2x_1$ 
```

```
(%o5)  $\frac{d}{dt}x_2 = 0.1x_1 - 0.4x_2$ 
```

Ara introduïm les condicions inicials, els punts inicial i final de l'interval en el qual es vol obtenir la solució, el pas o amplitud dels intervals en que es divideix l'interval anterior i es calcula em nombre d'intervals:

```
(%i6) x1_0:1;      x2_0:3;
      t_ini=t0:0.0;  t_fin=t1:10.0;      Pas=p:0.5;  n_int=n_int:(t1 - t0)/0.5;
(%o6) 1
      (%o7) 3
      (%o8) t_ini = 0.0
      (%o9) t_fin = 10.0
      (%o10) Pas = 0.5
      (%o11) n_int = 20.0
```

Per obtenir la solució establim ara una precisió numèrica dels resultats de wxMaxima:

```
(%i2) fpprintprec : 5 $
```

Ara calcularem la solució particular amb el mètode de Runge-Kutta, per a la qual cosa cal tenir de forma explícita la derivada de cada una de les funcions, que en aquest cas és la forma en que s'ha expressat el sistema. Vegem el resultat:


```
(%i13) solucio_LV: rk([ - 0.2*x1 + 0.3*x2 , 0.1*x1 - 0.4*x2], [x1, x2], [x1_0, x2_0], [t, t0, t1, p]);
(%o13) [[0.0, 1, 3], [0.5, 1.2961, 2.5088], [1.0, 1.5014, 2.1179], [1.5, 1.6374, 1.8055], [2.0, 1.7204, 1.5545], [
2.5, 1.7634, 1.3518], [3.0, 1.7762, 1.1871], [3.5, 1.7665, 1.0523], [4.0, 1.7403, 0.941], [4.5, 1.7021, 0.848], [5.0,
1.6554, 0.771], [5.5, 1.603, 0.705], [6.0, 1.5468, 0.648], [6.5, 1.4886, 0.6], [7.0, 1.4294, 0.557], [7.5, 1.3701,
0.519], [8.0, 1.3113, 0.486], [8.5, 1.2537, 0.456], [9.0, 1.1975, 0.429], [9.5, 1.1429, 0.404], [10.0, 1.0902, 0.38
]]
```

La solució es dona en forma de llista amb les ternes de valors que corresponen a la variable independent i a cada una de les variables dependents. Això obliga a fer algunes operacions per tal d'obtenir les llistes corresponents a cada una de les variables. Els detalls de la sintaxi es justificaran en l'Apèndix dedicat a Àlgebra Lineal.

```
(%i14) M1:matrix(
      [0.0,1,3],[0.5,1.2961,2.5088],[1.0,1.5014,2.1179],[1.5,1.6374,1.8055],[2.0,1.7204,1.5545],
      [2.5,1.7634,1.3518],[3.0,1.7762,1.1871],[3.5,1.7665,1.0523],[4.0,1.7403,0.941],
      [4.5,1.7021,0.848],[5.0,1.6554,0.771],[5.5,1.603,0.705],[6.0,1.5468,0.648],[6.5,1.4886,0.6],
      [7.0,1.4294,0.557],[7.5,1.3701,0.519],[8.0,1.3113,0.486],[8.5,1.2537,0.456],
      [9.0,1.1975,0.429],[9.5,1.1429,0.404],[10.0,1.0902,0.381]
      )$
(%i15) M1T:transpose(M1)$
      T:first(M1T)$      Y1:second(M1T)$      Y2:third(M1T)$
(%i19) List1:makelist([T[n],Y1[n]], n, 1, 1 + n_int)$
      List2:makelist([T[n],Y2[n]], n, 1, 1 + n_int)$
```

Ara es pot obtenir la representació gràfica de cada una de les llistes, és a dir, de cada una de les solucions del sistema:

```
(%i21) wxplot2d(
[ [discrete, List1] , [discrete, List2] ] ,
[style, [lines, 2, 1] , [lines, 2, 2] , [lines, 2, 3] ]
);
```



Vegem a continuació un clàssic dels sistemes d'EDO: les equacions de Lotka-Volterra que expressen la relació entre el nombre d'individus d'un ecosistema predador i presa. En aquest cas N_1 indica el nombre de preses (llebres) i N_2 indica el nombre de predadors (linxs). Els valors dels paràmetres de les equacions s'obtenen a partir de dades experimentals per regressió multivariant.

Exemple 12.4.4.- Es tracta de calcular la solució particular del sistema d'equacions diferencials:

$$\begin{cases} N_1' = a_1 N_1 - a_2 N_1 N_2 \\ N_2' = -b_1 N_2 + b_2 N_1 N_2 \end{cases}$$

amb les condicions inicials $N_1(0) = 30, N_2(0) = 6$.

Introduïm les equacions del sistema:

```
(%i4) edo1:'diff(N1, t)= a1*N1 - a2*N1*N2;
edo2:'diff(N2, t)= - b1*N2 + b2*N1*N2;
```

```
(%o4)  $\frac{d}{dt}N1 = a1 N1 - a2 N1 N2$ 
```

```
(%o5)  $\frac{d}{dt}N2 = b2 N1 N2 - b1 N2$ 
```

Ara introduïm les condicions inicials i els paràmetres de les equacions diferencials:

```
(%i6) N1_0:30;      N2_0:6;
      a1:0.5654;    a2:0.0245;    b1:0.8267;    b2:0.0219;
(%o6) 30
      (%o7) 6
      (%o8) 0.565
      (%o9) 0.0245
      (%o10) 0.827
      (%o11) 0.0219
```

Ara introduïm els punts inicial i final de l'interval en el qual es vol obtenir la solució, el pas o amplitud dels intervals en que es divideix l'interval anterior i es calcula em nombre d'intervals:

```
(%i12) t_ini=t0:0.0;    t_fin=t1:20.0;    Pas=p:0.5;    n_int=n_int:(t1 - t0)/0.5;
(%o12) t_ini= 0.0
      (%o13) t_fin= 20.0
      (%o14) Pas= 0.5
      (%o15) n_int= 40.0
```

Per obtenir la solució establim ara una precisió numèrica dels resultats de wxMaxima:

```
(%i2) fpprintprec : 5 $
```

Ara calcularem la solució particular amb el mètode de Runge-Kutta, per a la qual cosa cal tenir de forma explícita la derivada de cada una de les funcions, que en aquest cas és la forma en que s'ha expressat el sistema. Vegem el resultat:

```
(%i17) solucio_LV: rk([ a1*N1 - a2*N1*N2, - b1*N2 + b2*N1*N2], [N1, N2], [N1_0, N2_0], [t, t0, t1, p]);
(%o17) [[0.0, 30, 6], [0.5, 37.062, 5.721], [1.0, 45.805, 5.9473], [1.5, 56.232, 6.8679], [2.0, 67.82, 8.9548], [
2.5, 78.782, 13.243], [3.0, 84.883, 21.613], [3.5, 79.734, 35.646], [4.0, 61.802, 51.611], [4.5, 40.965, 59.755],
5.0, 26.386, 56.753], [5.5, 18.405, 47.734], [6.0, 14.477, 37.697], [6.5, 12.794, 28.905], [7.0, 12.455, 21.932],
7.5, 13.065, 16.669], [8.0, 14.488, 12.811], [8.5, 16.727, 10.046], [9.0, 19.871, 8.1114], [9.5, 24.073, 6.8174],
10.0, 29.533, 6.0399], [10.5, 36.474, 5.7259], [11.0, 45.087, 5.9101], [11.5, 55.396, 6.7669], [12.0, 66.937,
8.7396], [12.5, 78.055, 12.808], [13.0, 84.734, 20.793], [13.5, 80.615, 34.424], [14.0, 63.416, 50.564], [14.5,
42.366, 59.56], [15.0, 27.223, 57.265], [15.5, 18.835, 48.491], [16.0, 14.678, 38.423], [16.5, 12.867, 29.504], [
17.0, 12.445, 22.393], [17.5, 12.991, 17.011], [18.0, 14.355, 13.06], [18.5, 16.529, 10.222], [19.0, 19.601, 8.23],
[19.5, 23.717, 6.8958], [20.0, 29.075, 6.0823]]
```

La solució es dona en forma de llista amb les ternes de valors que corresponen a la variable independent i a cada una de les variables dependents. Això obliga a fer les operacions que s'han comentat en l'exemple anterior:

```
(%i3) M1:matrix(
[0.0,30.6],[0.5,37.062,5.721],[1.0,45.805,5.9473],[1.5,56.232,6.8679],[2.0,67.82,8.9548],
[2.5,78.782,13.243],[3.0,84.883,21.613],[3.5,79.734,35.646],[4.0,61.802,51.611],[4.5,40.965,59.755],
[5.0,26.386,56.753],[5.5,18.405,47.734],[6.0,14.477,37.697],[6.5,12.794,28.905],[7.0,12.455,21.932],
[7.5,13.065,16.669],[8.0,14.488,12.811],[8.5,16.727,10.046],[9.0,19.871,8.1114],[9.5,24.073,6.8174],
[10.0,29.533,6.0399],[10.5,36.474,5.7259],[11.0,45.087,5.9101],[11.5,55.396,6.7669],[12.0,66.937,8.7396],
[12.5,78.055,12.808],[13.0,84.734,20.793],[13.5,80.615,34.424],[14.0,63.416,50.564],[14.5,42.366,59.56],
[15.0,27.223,57.265],[15.5,18.835,48.491],[16.0,14.678,38.423],[16.5,12.867,29.504],[17.0,12.445,22.393],
[17.5,12.991,17.011],[18.0,14.355,13.06],[18.5,16.529,10.222],[19.0,19.601,8.2328],[19.5,23.717,6.8958],
[20.0,29.075,6.0823]
)$

(%i19) M1T:transpose(M1)$
T:first(M1T)$ Y1:second(M1T)$ Y2:third(M1T)$ M:Y1+Y2$

(%i24) List1:makelist([T[n],Y1[n]], n, 1, 1 + n_int)$
List2:makelist([T[n],Y2[n]], n, 1, 1 + n_int)$
List3:makelist([T[n],M[n]], n, 1, 1 + n_int)$
```

Ara es pot obtenir la representació gràfica de cada una de les llistes, és a dir, de cada una de les solucions del sistema:

```
(%i27) wxplot2d(
[ [discrete, List1], [discrete, List2], [discrete, List3] ],
[style, [lines, 2, 1], [lines, 2, 2], [lines, 2, 3] ]
);
```

